Laplace Transforms

We need Maple's integral transform package inttrans for the Laplace transform.

```
> restart;with(inttrans);
```

$[addtable, fourier, fouriercos, fouriersin, hankel, hilbert, invfourier, invhilbert, invlaplace, invmellin,$
$laplace, mellin, savetable]$

Laplace Transform

The Laplace transform of a function $f(t)$ is given by the formula $F(s) = \boldsymbol{L}\{f(t)\} = \int_0^\infty \mathrm{e}^{-st} f(t)\, \mathrm{d}t$ for all values of s for which the improper integral converges.

The command for finding the Laplace transform $F(s)$ of a function $f(t)$ is **laplace(f(t),t,s)**. Let's look at some examples.

$f(t) = \mathrm{e}^{-b\,t}$.

```
> laplace(exp(b*t),t,s);
```

$$\frac{1}{s-b}$$

$f(t) = 11 + 5\,\mathrm{e}^{4\,t} - 6\sin(2\,t)$.

```
> laplace(11+5*exp(4*t)-6*sin(2*t),t,s);
```

$$\frac{11}{s} + \frac{5}{s-4} - \frac{12}{s^2+4}$$

Unit Step (Heaviside) and Dirac delta Functions

We will begin by plotting Heaviside $(t$ - 4).

```
> plot(Heaviside(t-4),t=-10..10, discont=true, thickness=3);
```



This function is undefined at 4, but takes a one unit step as 4 is crossed.

```
> "Heaviside(3.99)"=Heaviside(3.99-4);
  "Heaviside(4)"=Heaviside(4-4);
  "Heaviside(4.01)"=Heaviside(4.01-4);
```

$$\text{"Heaviside(3.99)"} = 0.$$

$$\text{"Heaviside(4)"} = \mathit{undefined}$$

$$\text{"Heaviside(4.01)"} = 1.$$

Next we look at the Laplace transform of $u(t-b)$.

```
> laplace(Heaviside(t-b),t,s);
```

$$\mathit{laplace}(\mathrm{Heaviside}(t-b), t, s)$$

We substitute $b = 4$ and simplify.

```
> subs(b=4,%);
```

$$\mathit{laplace}(\mathrm{Heaviside}(t-4), t, s)$$

```
> simplify(%);
```

$$\frac{e^{-4s}}{s}$$

We will begin thinking of the Dirac delta function as the derivative of the unit step function, i.e.

$\text{delta}(t-b) = \frac{\partial}{\partial t} u(t-b).$

```
> diff(Heaviside(t-b),t);
```

$$\text{Dirac}(t-b)$$

Let's plot $\delta(t - 4)$.

```
> plot(Dirac(t-4),t=-10..10, y=-1..1, discont=true, thickness=3);
```



From the graph, it seems clear that $\text{delta}(t-b) = 0$ everywhere except at $t = b$ where it is undefined. But let's look at some of it's integrals when $b = 4$.

```
> Int(Dirac(t-4),t=-infinity..3.9999999999)=int(Dirac(t-4),t=-
  infinity..3.9999999999);
```

$$\int_{-\infty}^{3.9999999999} \text{Dirac}(t-4)\, dt = 0.$$

```
> Int(Dirac(t-4),t=4.0000000001..infinity)=int(Dirac(t-4),t=
  4.0000000001..infinity);
```

$$\int_{4.0000000001}^{\infty} \text{Dirac}(t-4)\, dt = 0.$$

```
> Int(Dirac(t-4),t=3.9999999999..4.0000000001)=int(Dirac(t-4),t=
  3.9999999999..4.0000000001);
```

$$\int_{3.9999999999}^{4.0000000001} \mathrm{Dirac}(t-4)\, dt = 1.000000000$$

```
> Int(Dirac(t-4),t=-infinity..4)=int(Dirac(t-4),t=-infinity..4);
```

$$\int_{-\infty}^{4} \mathrm{Dirac}(t-4)\, dt = \frac{1}{2}$$

```
> Int(Dirac(t-4),t=4..infinity)=int(Dirac(t-4),t=4..infinity);
```

$$\int_{4}^{\infty} \mathrm{Dirac}(t-4)\, dt = \frac{1}{2}$$

It appears that $\int_a^b \mathrm{Dirac}(t-4)\, dt$ is 0 for any interval that does not contain 4 between the limits of integration and 1 for any interval that does. However, if 4 is a limit of integration, the integral is $\frac{1}{2}$. From the area interpretation of the integral, it appears that all of the area is being acquired precisely at $t = 4$, the point where the function is undefined. Let's look at a plot of $\int_{-\infty}^{t} \mathrm{Dirac}(\tau-4)\, d\tau$.

```
> plot(int(Dirac(tau-4),tau=-infinity..t),t=-10..10,discont=true,
  thickness=3);
```



But this is the plot of the unit step function. This makes some sense since if the Dirac delta function is the derivative of the unit step function, then the unit step function is the integral of the Dirac delta function. What is happening here is that the Dirac delta function is not really a function in the usual sense, but something called a **distribution**. Although it is undefined at $t = b$, we think of $\mathrm{delta}(4) = \infty$. Physically, it can be interpreted as the moment in time when an impulsive force such as a hammer blow is applied or a switch turned on Let's look at its Laplace transform, where we must assume $b > 0$.

```
> assume(b>0);
> laplace(Dirac(t-b),t,s);
```

$$e^{-s\,b\sim}$$

```
> laplace(Dirac(t-4),t,s);
```

$$e^{-4\,s}$$

The following removes the assumption on b.

```
> b:='b';
```

$$b := b$$